

```

;=====
; A RUDIMENTARY MUSIC PLAYER PROGRAM FOR WICHIT'S 8080 DEV BOARD
;       v3.3 by DeviceGuru (http://deviceguru.com)
; NOTES...
; 1. this was inspired by the listing here:
; http://kevindriscoll.org/projects/ccswg2012/fool\_on\_a\_hill.html
; 2. this was compiled with zasm, available for free download here:
; http://k1.spdns.de/Develop/Projects/zasm/Distributions/
; or via online compiler at: http://k1.spdns.de/cgi-bin/zasm.cgi
; 3. Be sure to set the 8080 board's interrupt switch to "10ms tick"
;=====
;
#target bin
.8080
#code FILLER,0,0x8100 ; create offset so code compiles to begin at 8100h in the
board's RAM
#code PROGRAM
prep:                ; initialize interrupt vector
    mvi a,$c3
    sta 8038h
    mvi a,$12
    sta 8039h
    mvi a,$81
    sta 803ah
    jmp begin        ; jump around ISR
; =====
interrupt_service_routine:
    dcr c            ; this creates note tempo
    ei               ; decrement C and return
    ret
; =====
begin:
    lxi h,daisy      ; point to notecode table for desired song
get_note:
    mov a,m           ; get current notecode
    out $00           ; show notecode on LEDs
    cpi $ff           ; is note an ff?
    jz done           ; if yes we're done
    mov a,m           ; get notecode again
    cpi $00           ; is note a 00?
    jz play_rest      ; if yes go do 1-beat rest
    jmp play_note     ; if no go do 1-beat note
;
done:
    mvi a,$aa
    out $00           ; show end code in LEDs
    di               ; disable interrupts
    RST 3             ; done!
;
; =====
;
play_note:
    mvi c,tempo       ; set tempo (1 beat ~ tempo*20 ms)
    ei               ; enable interrupts
; do first half of square wave
noteloop:
    mvi a,$7f         ; turn speaker DC on
    out $40

```

```

    mov a,m          ; get notecode again
;
loop1:
    dcr a
    jnz loop1        ; continue for 1st half of square wave
;
    mvi a,$ff        ; turn speaker DC off
    out $40
    mov a,m          ; get notecode again
loop2:
    dcr a
    jnz loop2        ; loop to complete 2nd half of square wave
;
    mov a,c          ; see if interrupt-decremented duration is completed
    cpi $00
    jnz noteloop     ; continue note for one beat
;
    inx h            ; point to next note
    jmp get_note
;
play_rest:
    mvi c,tempo      ; set tempo (1 beat = tempo*20 ms)
    ei
loop3:
    mov a,c          ; loop for one beat
    cpi $00
    jnz loop3        ; continue rest for one beat
    inx h            ; point to next note
    jmp get_note
;
; =====
;
tempo equ $25        ; quarter note duration at moderato rate
;
;daisy: ; notes for Daisy Daisy song...
    db n_c5,n_c5,n_c5,n_a4,n_a4,n_a4,n_f4,n_f4,n_f4,n_c4,n_c4,n_c4
;    dai-          sy          dai-          sy
    db n_d4,n_e4,n_f4,n_d4,n_d4,n_f4,n_c4,n_c4,n_c4,n_c4,n_c4,n_c4
;    give me your an-      swer do
    db n_g4,n_g4,n_g4,n_c5,n_c5,n_c5,n_a4,n_a4,n_a4,n_f4,n_f4,n_f4
;    i'm          half      cra          sy
    db n_d4,n_e4,n_f4,n_g4,n_g4,n_a4,n_g4,n_g4,n_g4,n_g4,n_g4,n_a4
;    all for the love of you      it
    db n_b4f,n_a4,n_g4,n_c5,n_c5,n_a4,n_g4,n_f4,n_f4,n_f4,n_f4,n_g4
;    won't be a sty lish mar riage,      I
    db n_a4,n_a4,n_f4,n_d4,n_d4,n_f4,n_d4,n_c4,n_c4,n_c4,n_c4,n_c4
;    can't a ford the car riage,      but
    db n_f4,n_f4,n_a4,n_g4,$00,$00,n_f4,n_f4,n_a4,n_g4,n_a4,n_b4f
;    you'll look sweet,      on the seat of a
    db n_c5,n_a4,n_f4,n_g4,n_g4,n_c4,n_f4,n_f4,n_f4,n_f4,$ff
;    by- cy- cle built for two
;
sunshine: ; notes for You Are My Sunshine song...
    db $00,n_g3,n_c4,n_d4,n_e4,n_e4,n_e4,n_e4
;    (rest) you are my sun -shine
    db $00,n_e4,n_d4,n_e4,n_c4,n_c4,n_c4,n_c4
;    (rest) my on- ly sun- shine
    db $00,n_c4,n_d4,n_e4,n_f4,n_f4,n_a4,n_a4
;    (rest) you make me hap- py

```

```

    db $00,n_a4,n_g4,n_f4,n_e4,n_e4,n_e4,n_e4
;    (rest) when skys are gray
    db $00,n_c4,n_d4,n_e4,n_f4,n_f4,n_a4,n_a4
;    (rest) You'll ne- ver know      dear,
    db $00,n_a4,n_g4,n_f4,n_e4,n_e4,n_c4,n_c4
;    (rest) how much I love      you
    db $00,$00,n_c4,n_d4,n_e4,n_e4,n_e4,n_f4
;    (rest)(rest) please don't take my
    db n_d4,n_d4,n_d4,n_e4,n_c4,n_c4,n_c4,n_c4,$ff
;    sun-      shine a- way
;
birthday: ; notes for Happy Birthday song...
    db n_c4,n_c4,n_d4,n_d4,n_c4,n_c4,n_f4,n_f4,n_e4,n_e4,n_e4,$00
;    ha- py    birth    -day      to      you
    db n_c4,n_c4,n_d4,n_d4,n_c4,n_c4,n_g4,n_g4,n_f4,n_f4,n_f4,$00
;    ha- py    birth    -day      to      you
    db n_c4,n_c4,n_c5,n_c5,n_a4,n_a4,n_f4,n_f4,n_e4,n_e4,n_d4,n_d4,$00,$00
;    ha- py    birth    -day      dear      ha-      zel
    db n_b4f,n_b4f,n_a4,n_a4,n_f4,n_f4,n_g4,n_g4,n_f4,n_f4,n_f4,$ff
;    ha- py    birth    -day      to      you!
;
; === Note codes from c3 through c5 ===
;n_c3 equ $ ; 131 Hz
;n_d3 equ $ ; 147 Hz
;n_e3 equ $ ; 165 Hz
;n_f3 equ $ ; 175 Hz
n_g3 equ $f7 ; 185 Hz
n_a3 equ $dc ; 220 Hz
n_b3f equ $d0 ; 233 Hz
n_b3 equ $c6 ; 247 Hz
n_c4 equ $b9 ; 262 Hz
n_d4 equ $a5 ; 294 Hz
n_e4 equ $93 ; 330 Hz
n_f4 equ $8b ; 349 Hz
n_g4 equ $7b ; 392 Hz
n_a4 equ $6e ; 440 Hz
n_b4f equ $68 ; 466 Hz
n_b4 equ $62 ; 494 Hz
n_c5 equ $5d ; 523 Hz
n_d5 equ $52 ; 587 Hz
n_e5 equ $49 ; 660 Hz
n_f5 equ $45 ; 698 Hz
n_g5 equ $3e ; 784 Hz
n_a5 equ $37 ; 880 Hz
n_b5f equ $34 ; 932 Hz
n_b5 equ $31 ; 988 Hz
n_c6 equ $2e ; 1046 Hz
;=====
.end

```